

HEWLETT-PACKARD

HP-38E/38C

PERSONAL FINANCE

Applications



NOTICE

Hewlett-Packard Company makes no express or implied warranty with regard to the keystroke procedures and program material offered or their merchantability or other fitness for any particular purpose. The keystroke procedures and program material are made available solely on an "as is" basis, and the entire risk as to their quality and performance is with the user. Should the keystroke procedures or program material prove defective, the user (and not Hewlett-Packard Company nor any other party) shall bear the entire cost of all necessary correction and all incidental or consequential damages. Hewlett-Packard Company shall not be liable for any incidental or consequential damages in connection with or arising out of the furnishing, use, or performance of the keystroke procedures or program material.



HP-38E/38C

**Personal Finance
Applications**

January 1982

00038-90052 Rev. B

Introduction

This *Personal Finance Applications* book has been designed to supplement the HP-38E/38C *Owner's Handbook* by providing a collection of applications specifically for personal finance. Programs and/or step by step keystroke procedures with corresponding examples for 10 problem types are explained. Hopefully, this book will provide a reference guide to many of your problems, and show you how to redesign our examples to fit your specific needs.

All results are carried internally to ten significant digits. If intermediate answers are rounded by the user, slightly different final values may be obtained.

Contents

Introduction	2
Table of Contents	3
Programs	
Tax Free Individual Retirement Account (IRA) or Keogh Plan	4
Stock Portfolio Evaluation and Analysis	8
U.S. Treasury Bill Valuation	12
True Annual Growth Rate of an Investment Portfolio	14
Bond Purchased Between Coupons	19
The True Cost of an Insurance Policy	22
Real Estate Equity Investment Analysis	25
Homeowner's Monthly Payment Estimator	30
True Annual Percentage Interest Rate on a Mortgage With Fees	33
Rent Versus Buy	36
Appendix	39

Tax Free Individual Retirement (IRA) or Keogh Plan

The advent of tax free retirement accounts (IRA or Keogh) has resulted in considerable benefits for many persons who are not able to participate in group profit sharing or retirement plans. The savings due to tax free status are often considerable, but complex to calculate. Required data are: the years to retirement, the total annual investment, the compound annual interest rate of the investment and an assumed tax rate which would be paid on a similar non tax-free investment. This program calculates:

1. The future cash value of the “tax-free” investment.
2. The total cash paid in.
3. The total dividends paid.
4. The future value of the investment at retirement, assuming that after retirement you withdrew the money at a rate which causes the money to be taxed at $\frac{1}{2}$ the rate at which it would otherwise have been taxed during the pay-in period.
5. The diminished purchasing power assuming a given annual inflation rate.
6. The future value of a comparable taxable investment.
7. The diminished purchasing power of 6).

Notes:

- The calculations run from the beginning of the first year to the end of the last year.
- The interest (annual yield), i , should be entered to as many significant figures as possible for maximum accuracy.
- The assumed 7% annual inflation rate may be changed by modifying the program at line 20 (or, if over 9%, at lines 19 and 20).
- The assumed tax rate used to calculate the after tax value of the “tax free” investment may be changed by modifying the program at line 09.

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
g P/R g CL P	00-	7	20- 7
RCL n	01- 22 11	ENTER	21- 31
RCL PMT	02- 22 14	RCL n	22- 22 11
x	03- 61	g y^x	23- 25 21
R/S	04- 74	÷	24- 71
+	05- 51	R/S	25- 74
R/S	06- 74	RCL i	26- 22 12
RCL 3	07- 22 3	1	27- 1
.	08- 73	RCL 3	28- 22 3
5	09- 5	%	29- 23
%	10- 23	-	30- 41
CHS	11- 32	x	31- 61
1	12- 1	i	32- 12
+	13- 51	FV	33- 15
RCL FV	14- 22 15	g LAST x	34- 25 31
x	15- 61	x	35- 61
R/S	16- 74	R/S	36- 74
1	17- 1	g GTO 17	37- 25 7 17
.	18- 73	g P/R	
0	19- 0		

REGISTERS			
R ₀	R ₁	R ₂	R ₃ tax %

1. Key in the program.
2. Set the payment switch to BEGIN; press **f** **CLEAR** **ALL**.
3. Key in the tax rate as a percent and press **STO** 3.
4. Key in years to retirement and press **n**.

6 Tax Free IRA or Keogh Plan

5. Key in the interest rate as a percent and press **[i]**.
6. Key in the annual payment and press **[CHS] [PMT]**.
7. Press **[FV]** to calculate the future value of the “tax free” investment.
8. Press **[R/S]** to compute the total cash paid in.
9. Press **[R/S]** to compute the total dividends paid.
10. Press **[R/S]** to compute the future value when, after retirement, money is withdrawn at a rate causing the tax rate to equal $\frac{1}{2}$ the rate paid during the pay-in period.
11. Press **[R/S]** to compute the diminished purchasing power, in terms of today’s dollars, of the future value assuming a 7% annual inflation rate.
12. Press **[R/S]** to compute the future value of an ordinary tax investment.
13. Press **[R/S]** to compute the diminished purchasing power of the above amount.

Example:

Assuming a 35 year investment period with a dividend rate of 8.175% and a tax rate of 40%:

1. If you invest \$1500 each year in a tax free account, what will its value be at retirement?
2. How much cash will be paid in?
3. What will the value of the earned dividends be?
4. After retirement, if you withdraw cash from the account at a rate such that it will be taxed at a rate equal to one-half the rate paid during the pay-in period, what will be the after tax value?
5. What is the diminished purchasing power of that amount, in today’s dollars, assuming 7% annual inflation?
6. If you invest the same amount (\$1500) * each year with dividends taxed as ordinary income, what will the total tax-paid cash at retirement be?
7. What is the purchasing power of that figure in terms of today’s dollars?

* **Note:** \$1500 *after taxes* for a non-Keogh or IRA account.

KeystrokesBEGIN  END CLEAR 40  335 8.175 1500  **Display****0.00****40.00****35.00****8.18****-1,500.00****Keystrokes****Display****290,730.34****-52,500.00****238,230.34****232,584.27****21,784.53****83,616.05****7,831.73**

(1)

(2)

(3)

(4)

(5)

(6)

(7)

Stock Portfolio Evaluation and Analysis

This program evaluates a portfolio of stocks given the current market price per share and the annual dividend. The user inputs the initial purchase price of a stock, the number of shares, the beta coefficient *, the annual dividend, and the current market price for a portfolio of any size.

The program returns the percent change in value of each stock and the valuation and beta coefficient * of the entire portfolio. Output includes the original portfolio value, the new portfolio value, the percent change in value and the annual dividend and yield as a percent of the current market value. The overall beta coefficient of the portfolio is also calculated.

Notes:

- Prices are input in the form XXX.ND where N is the numerator and D the denominator of the fractional portion of the price, e.g., 25 $\frac{58}{100}$ is input as 25.58.
- The beta coefficient analysis is optional. Just key in 1.00 if beta is not to be analyzed.

* The beta coefficient is a measure of a stock variability (risk) compared to the market in general. Beta values for individual stocks can be acquired from brokers, investment publications or the local business library.

KEY ENTRY	DISPLAY
<input type="button" value="9"/> <input type="button" value="P/R"/> <input type="button" value="9"/> <input type="button" value="CL P"/>	00-
<input type="button" value="STO"/> 7	01- 21 7
<input type="button" value="9"/> <input type="button" value="FRAC"/>	02- 25 61
<input type="button" value="9"/> <input type="button" value="X=0"/>	03- 25 6
<input type="button" value="9"/> <input type="button" value="GTO"/> 15	04- 25 7 15
1	05- 1
0	06- 0
<input type="button" value="x"/>	07- 61

KEY ENTRY	DISPLAY
<input type="button" value="f"/> <input type="button" value="INTGR"/>	08- 24 61
<input type="button" value="9"/> <input type="button" value="LAST X"/>	09- 25 31
<input type="button" value="9"/> <input type="button" value="FRAC"/>	10- 25 61
<input type="button" value="÷"/>	11- 71
1	12- 1
0	13- 0
<input type="button" value="÷"/>	14- 71
<input type="button" value="RCL"/> 7	15- 22 7

KEY ENTRY	DISPLAY
f INTGR	16- 24 61
+	17- 51
RCL 0	18- 22 0
g x=0	19- 25 6
g GTO 38	20- 25 7 38
STO - 0	21- 21 41 0
x	22- 61
RCL 1	23- 22 1
x	24- 61
STO + 2	25- 21 51 2
x y	26- 33
RCL 1	27- 22 1
x	28- 61
STO + 3	29- 21 51 3
g R+	30- 25 33
x	31- 61
STO + 6	32- 21 51 6
RCL 4	33- 22 4
g LAST X	34- 25 31
f Δ%	35- 24 23
R/S	36- 74
g GTO 01	37- 25 7 01
+	38- 51
x y	39- 33

KEY ENTRY	DISPLAY
STO 1	40- 21 1
x	41- 61
STO 4	42- 21 4
STO + 5	43- 21 51 5
1	44- 1
STO 0	45- 21 0
R/S	46- 74
g GTO 01	47- 25 7 01
RCL 5	48- 22 5
R/S	49- 74
RCL 2	50- 22 2
R/S	51- 74
f Δ%	52- 24 23
R/S	53- 74
RCL 2	54- 22 2
RCL 3	55- 22 3
R/S	56- 74
f %T	57- 24 22
R/S	58- 74
RCL 6	59- 22 6
RCL 2	60- 22 2
÷	61- 71
g GTO 00	62- 25 7 00
g P/R	

REGISTERS			
R ₀ Flag	R ₁ S _i	R ₂ ΣPV	R ₃ Σ Div.
R ₄ P _i	R ₅ Σ Orig. Val.	R ₆ Σ P _i S _i P _i	R ₇ CCC.ND

10 Stock Portfolio Evaluation and Analysis

1. Key in the program.
2. Initialize the program by pressing **f** CLEAR **ALL**.
3. Key in the number of shares of a stock and press **ENTER**.
4. Key in the initial purchase price of the stock and press **R/S**.
5. Key in the beta coefficient of the stock and press **ENTER**.
6. Key in the annual dividend of the stock and press **ENTER**.
7. Key in the present price of the stock and press **R/S**, the display will show the percent change in the stock value.
8. Repeat steps 3 through 7 until all the stocks are entered.
9. Next, to evaluate the entire portfolio, press **g** **GTO** 48. Then:
10. Press **R/S** to see the initial portfolio value.
11. Press **R/S** to see the present portfolio value.
12. Press **R/S** to see the percent change in value.
13. Press **R/S** to see the total yearly dividend.
14. Press **R/S** to see the annual dividend yield as a percent of the current market value.
15. Press **R/S** to see the beta coefficient of the portfolio.

Example:

Evaluate the following portfolio:

number of shares held	initial purchase price	beta coefficient	annual dividend	present market price	stock
100	25 $\frac{1}{8}$.8	\$1.70	27 $\frac{1}{4}$	Int'l Heartburn
200	30 $\frac{1}{4}$	1.2	\$2.10	33 $\frac{1}{2}$	P.D.Q.
50	89 $\frac{7}{8}$	1.3	\$4.55	96 $\frac{1}{8}$	Datacrunch
500	65 $\frac{1}{4}$.6	\$3.50	64 $\frac{3}{8}$	N.W. Sundial

Keystrokes

Display

f CLEAR ALL	0.00
100 ENTER	100.00
25.58 R/S	1.00
.8 ENTER	0.80
1.70 ENTER	1.70
27.14 R/S	6.34

% change in stock's value

Keystroke:

200	ENTER	200.00	
30.14	R/S	1.00	
1.2	ENTER	1.20	
2.10	ENTER	2.10	
33.12	R/S	10.74	% change in stock's value
50	ENTER	50.00	
89.78	R/S	1.00	
1.3	ENTER	1.30	
4.55	ENTER	4.55	
96.18	R/S	6.95	% change in stock's value
500	ENTER	500.00	
65.14	R/S	1.00	
.6	ENTER	0.60	
3.50	ENTER	3.50	
64.38	R/S	-1.34	% change in stock's value
g GTO	48		
R/S		45,731.25	Original value
R/S		46,418.75	Present value
R/S		1.50	% change in value
R/S		2,567.50	Total yearly dividend
R/S		5.53	Yearly dividend yield
R/S		0.77	Portfolio β coefficient

U.S. Treasury Bill Valuation

This HP-38E/38C program calculates the price per \$100 and the dollar value of U.S. Treasury Bills given the face amount, the quote and maturity dates and the bid and ask quotations (as a percent yield). The price per \$100 and the dollar value may be calculated on either bid or ask quotation alone if only one is known.

Notes:

- Dates must be entered in the proper format, including the year. Be sure Date Format switch is set to match your input format (see Page 24 of the HP-38E/38C Owner's Handbook).
- Provision is automatically made for leap years.
- Program limits days to maturity to a maximum of 360 in accordance with standard practice.

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
g P/R g CL P	00-	g EEX	11- 25 32
f ΔDAYS	01- 24 41	2	12- 2
R/S	02- 74	-	13- 41
+	03- 51	CHS	14- 32
2	04- 2	R/S	15- 74
÷	05- 71	g LAST X	16- 25 31
×	06- 61	÷	17- 71
3	07- 3	×	18- 61
6	08- 6	g GTO 00	19- 25 7 00
0	09- 0	g P/R	
÷	10- 71		

To use the program:

1. Key in the program and set the Date Format switch to the proper setting.

2. Key in the quotation date, press **ENTER** . Key in the due date, press **R/S** . The number of days between dates is displayed.
3. Key in the bid quotation, press **ENTER** . Key in the ask quotation and press **R/S** to calculate the price per \$100. (If only the bid or ask quotation is known, key it in, press **ENTER** **R/S** to use the single quote to calculate the price.)
4. Key in the face value, press **R/S** to calculate the market value.
5. For another bill, go to step 2.

Example:

The following dates and quotations were given for three treasury bills. Find the prices per \$100 and the dollar value of each.

Face Amount	Quote Date	Due Date	Bid	Ask
\$100,000	Feb. 5, 1979	May 15, 1979	5.75%	5.5%
\$ 50,000	Nov. 15, 1979	March 20, 1980	5.5%	5.25%
\$ 70,000	Dec. 10, 1979	Jan. 15, 1980	—	5.0%

Keystrokes:

Display:

D.MY  M.DY

2.051979	ENTER		
5.151979	R/S	99.00	Days between dates
5.75	ENTER 5.5	R/S 98.45	Price per \$100
100000	R/S	98,453.13	Dollar value
11.151979	ENTER		
3.201980	R/S	126.00	Days between dates
5.5	ENTER 5.25	R/S 98.12	Price per \$100
50000	R/S	49,059.38	Dollar value
12.101979	ENTER		
1.151980	R/S	36.00	Days between dates
5	ENTER R/S	99.50	Price per \$100
70000	R/S	69,650.00	Dollar value

True Annual Growth Rate of an Investment Portfolio

Finding the true rate of return on an investment portfolio where variable amounts are invested or withdrawn at indefinite or irregular intervals is a complicated calculation. However, it is a highly useful tool in money management.

This program uses ‘Fisher’s algorithm’, a powerful technique to solve for the true realized rate of return.

Notes:

- The program can accomodate a maximum of nine transactions.
- All transactions must be input in succession, oldest to newest.
- Enter only whole dollars (no cents) for P_j .
- For problems of this type, more than one, or no solutions may exist. The program is written to allow you to watch the solution converge, that is, program execution pauses to display the differences between successive approximations to the true interest rate. If this quantity does not get smaller as execution continues, no solution may exist. You may continue searching for a solution or try another solution by using a different initial guess for i (program instructions, step 5).
- The financial registers are used by this program to store miscellaneous data. Do not use them for other purposes when running the program.

Reference:

- 1) Lawrence Fisher, ‘*An Algorithm for finding Exact Rates of Return*’, The JOURNAL OF BUSINESS, University of Chicago Press, Volume XXXIX No. 1 part 2, January 1966.

KEY ENTRY	DISPLAY
g P/R g CL P	00-
RCL PMT	01- 22 14
f ΔDAYS	02- 24 41
3	03- 3
6	04- 6
5	05- 5
0	06- 0
0	07- 0
÷	08- 71
RCL PV	09- 22 13
g X<Y	10- 25 5
g GTO 15	11- 25 7 15
+	12- 51
g CF₁	13- 25 14
g GTO 00	14- 25 7 00
X<Y	15- 33
-	16- 41
g GTO 13	17- 25 7 13
1	18- 1
0	19- 0
n	20- 11
0	21- 0
STO 0	22- 21 0
STO 1	23- 21 1
RCL g CF₁	24- 22 25 14
g X=0	25- 25 6
g GTO 24	26- 25 7 24
f INTGR	27- 24 61
PV	28- 13

KEY ENTRY	DISPLAY
g LAST x	29- 25 31
g FRAC	30- 25 61
2	31- 2
g y^x	32- 25 21
f √x	33- 24 21
g EEX	34- 25 32
2	35- 2
x	36- 61
ENTER +	37- 31
ENTER +	38- 31
RCL i	39- 22 12
x	40- 61
g e^x	41- 25 22
RCL PV	42- 22 13
x	43- 61
STO + 0	44- 21 51 0
x	45- 61
STO + 1	46- 21 51 1
RCL n	47- 22 11
3	48- 3
g X<Y	49- 25 5
g GTO 24	50- 25 7 24
RCL i	51- 22 12
RCL FV	52- 22 15
RCL 0	53- 22 0
-	54- 41
RCL 1	55- 22 1
÷	56- 71
+	57- 51

16 True Annual Growth Rate

KEY ENTRY	DISPLAY
i	58- 12
g LAST X	59- 25 31
g PSE	60- 25 4
2	61- 2
g y^x	62- 25 21
f √x	63- 24 21
g EE_X	64- 25 32
7	65- 7

KEY ENTRY	DISPLAY
CHS	66- 32
g x≤y	67- 25 5
g GTO 18	68- 25 7 18
1	69- 1
RCL i	70- 22 12
f %T	71- 24 22
g P/R	

REGISTERS			
R ₀ Σ numerator	R ₁ Σ denominator	R ₂ CF ₁	R ₃ CF ₂
R ₄ CF ₃	R ₅ CF ₄	R ₆ CF ₅	R ₇ CF ₆
R ₈ CF ₇	R ₉ CF ₈	R ₀ CF ₉	R _n n
R _i i _k	R _{PV} T	R _{PMT} p _j	R _{FV} V _T

Instructions:

1. Key in the program.
2. Set switch to M.DY and press **f** **CLEAR** **ALL**.
3. Key in the date of the final fund evaluation and press **PMT**.
4. Key in the final fund value and press **FV**.
5. Key in an initial guess for the interest rate, .1, and press **i**.
6. Key in 2, press **n**.
7. Key in the transaction (whole dollars only) and press **PV**.
8. Key in the date of that transaction and press **R/S**.
9. Repeat steps 7 and 8 until all transactions are entered. Then press **g** **GTO** 18 **R/S** to compute the continuously compounded interest rate.
10. To compute the annual rate, use the following keystroke sequence: **RCL** **i** 1 **+** **g** **LN** 100 **x**.

Example:

You invest in a pension fund and after a number of years you are told that the fund value was 9,050 dollars on 7/01/1977. Determine the effective (continuously compounded) and nominal rate of return from the record of your transactions which are as follows:

- an investment of \$4,150 on 1/01/1972;
- an investment of \$2,005 on 1/01/1973;
- a withdrawal of \$950 on 1/01/1974;
- a withdrawal of \$3,111 on 1/01/1975;
- an investment of \$2,000 on 10/01/1975;
- an investment of \$2,005 on 1/01/1976;
- a final investment of \$1,000 On 7/01/1976.

Keystrokes**Display**D.MY  M.DY

[f] CLEAR [ALL]

0.00

7.011977 [PMT]

7.01

9050 [FV]

9,050.00

.1 [i]

0.10

2 [n]

2.00

4150 [PV]

4,150.00

1.011972 [R/S]

4,150.06

2005 [PV]

2,005.00

1.011973 [R/S]

2,005.04

950 [CHS] [PV]

-950.00

1.011974 [R/S]

-950.03

3111 [CHS] [PV]

-3,111.00

1.011975 [R/S]

-3,111.02

2000 [PV]

2,000.00

10.011975 [R/S]

2,000.02

2005 [PV]

2,005.00

1.011976 [R/S]

2,005.01

1000 [PV]

1,000.00

7.011976 [R/S]

1,000.01

[g] [GTO] 18

[R/S]

(-.04)

(-0.004)

(-0.00005)

(-0.00000001)

5.90

Differences between successive approximations of the correct interest rate.

effective continuous rate

18 True Annual Growth Rate

Keystrokes

RCL i

1 +

9 LN

100 x

Display

0.06

1.06

0.06

5.73

Annual percentage rate

Bond Purchased Between Coupons

A bond is a contract to pay interest, usually semi-annually, at a given rate (coupon), and to pay the principal of the bond at some specified future date. The value or price of a bond is the present value of the coupon payments plus the present value of the principal or redemption value, at a given interest rate (yield).

In periods of high inflation, bonds may be of great interest to investors, because of their high yields. Bonds are frequently purchased *between* coupon dates, making determination of their price and interest quite complicated.

In such cases, the Securities Industry Association has established certain formulae¹ to determine their price and yield. For semi-annual bonds held for more than 6 months, the following HP-38E/38C program evaluates bond price and accrued interest on an Actual/Actual day basis: For bonds calculated on a 30/360 day basis, two additional program steps are needed. Insert **[9] [R+]** after **[f] [ΔDAYS]** at steps 36 and 40.

¹Standard Securities Calculations Methods; Securities Industry Association; 1973.

KEY ENTRY	DISPLAY
[9] [P/R] [9] [CL P]	00-
[f] CLEAR [FIN]	01- 24 32
[RCL] 1	02- 22 1
[f] [INTGR]	03- 24 61
6	04- 6
.	05- 73
0	06- 0
1	07- 1
[9] [X≤Y]	08- 25 5
[9] [GTO] 21	09- 25 7 21
[RCL] 1	10- 22 1
5	11- 5

KEY ENTRY	DISPLAY
.	12- 73
9	13- 9
9	14- 9
9	15- 9
9	16- 9
9	17- 9
9	18- 9
[+]	19- 51
[9] [GTO] 24	20- 25 7 24
[RCL] 1	21- 22 1
6	22- 6
[-]	23- 41

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
STO 6	24- 21 6	RCL 7	41- 22 7
RCL 2	25- 22 2	±	42- 71
n	26- 11	n	43- 11
RCL 3	27- 22 3	0	44- 0
PMT	28- 14	PMT	45- 14
RCL 4	29- 22 4	FV	46- 15
i	30- 12	CHS	47- 32
RCL 5	31- 22 5	RCL n	48- 22 11
FV	32- 15	RCL 3	49- 22 3
PV	33- 13	CHS	50- 32
RCL 6	34- 22 6	x	51- 61
RCL 1	35- 22 1	R/S	52- 74
f △DAYS	36- 24 41	-	53- 41
STO 7	37- 21 7	g GTO 00	54- 25 7 00
RCL 6	38- 22 6	g P/R	
RCL 0	39- 22 0		
f △DAYS	40- 24 41		

REGISTERS			
R ₀ Settlement	R ₁ Next coupon	R ₂ # periods	R ₃ Coupon
R ₄ Yield	R ₅ Redemption	R ₆ Last coupon	R ₇ Used

1. Key in the program.
2. Set the Payment switch to END.
3. Key in the settlement (purchase) date (MM.DDYYYY); press **STO** 0.
4. Key in the date (MM.DDYYYY) of the next coupon; press **STO** 1.
5. Key in the *total* number of coupons which are received; press **STO** 2.
6. Key in the amount of the *semi-annual* coupon; press **STO** 3.*

7. Key in the *semi-annual* yield as a percent; press **[STO]** 4.
8. Key in the redemption value; press **[STO]** 5.*
9. Press **[R/S]** to obtain the amount of accrued interest (a share of the next coupon to which the seller is entitled).
10. Press **[R/S]** to determine the price of the bond.

Note: If a coupon is received on the last day of the month (i.e., October 31), an error condition may result. This happens because the count-back routine determines the last coupon date to be exactly 6 months earlier (i.e., April 31), and this may be an illegal date.

Example:

Given the following U.S. Treasury Bond, find its price:

Settlement date January 3, 1977; maturity date December 14, 1990 (28 coupon periods); next coupon date June 14, 1977; coupon rate 4.75%; yield 5%.

Keystrokes

BEGIN  END

1.031977	[STO] 0	1.03	Settlement date
6.141977	[STO] 1	6.14	Next coupon date
28	[STO] 2	28.00	Total number of coupons
4.75	[ENTER] 2 [÷]		
[STO] 3		2.38	Semi-annual coupon
5	[ENTER] 2 [÷]		
[STO] 4		2.50	Semi-annual yield
100	[STO] 5	100.00	Redemption value is not specified and is assumed to be 100
[R/S]		-0.26	Accrued interest
[R/S]		-97.51	Purchase price

To find bond yield, an iteration (trial and error) approach may be used with the bond price program. The user inputs successive “guesses” at the periodic yield into register 4 and solves for price. If the calculated price is the actual price paid, the yield is in register 4. If not, adjust the yield and repeat the procedure until the desired accuracy is obtained.

* Positive for cash received; negative for cash paid out.

The True Cost of an Insurance Policy

The true cost of an insurance policy other than term life insurance is rarely immediately apparent. The cost should include not only the premium payments but also the interest which could have been earned on the cash value or “savings portion” or the policy.

The following HP-38E/38C program calculates this true cost, (YPT_t) per \$1000 of protection and the interest rate paid on the savings elements of the policy.

Notes:

- To calculate YPT_t , you must assume some value for i ; for example, the interest rate you could earn on a one-year savings certificate, after tax.
- Similarly, to calculate i , you must assume a price per \$100 for alternative insurance; for example, a low cost term policy of the one-year renewable type.
- Even complex policies like minimum-deposit plans can be analyzed with this program. Use policy surrender values for cash-values, and the actual (after-tax) amounts for payments (premiums) and dividends.

References:

Joseph M. Belth, *Life Insurance—A Consumer's Handbook*, Indiana University Press, 1973, p. 234.

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
9 P/R 9 CL P	00-	9 EEEX	19- 25 32
ENTER +	01- 31	3	20- 3
1	02- 1	÷	21- 71
x ₂ y	03- 33	÷	22- 71
%	04- 23	R/S	23- 74
+	05- 51	9 LAST x	24- 25 31
RCL 1	06- 22 1	x	25- 61
RCL 2	07- 22 2	RCL 7	26- 22 7
+	08- 51	+	27- 51
STO 6	09- 21 6	RCL 6	28- 22 6
x	10- 61	÷	29- 71
RCL 3	11- 22 3	1	30- 1
RCL 4	12- 22 4	-	31- 41
+	13- 51	9 EEEX	32- 25 32
STO 7	14- 21 7	2	33- 2
-	15- 41	x	34- 61
RCL 5	16- 22 5	9 GTO 00	35- 25 7 00
RCL 3	17- 22 3	9 P/R	
-	18- 41		

REGISTERS

R ₀	R ₁ P _t	R ₂ V _{t-1}	R ₃ V _t
R ₄ D _t	R ₅ F _t	R ₆ Used	R ₇ Used

24 The True Cost of an Insurance Policy

1. Key in the program.
2. Initialize the program; press **f** CLEAR **ALL**.
3. Store the data:
Key in the annual premium for policy year t and press **STO** 1;
Key in the cash value at the beginning of the year and press **STO** 2.
Key in the cash value at the end of the year and press **STO** 3;
Key in the dividend for that year and press **STO** 4;
Key in the face amount of the policy and press **STO** 5.
4. To compute the true cost per \$1000 of protection in policy year t , key in the interest rate available to you elsewhere and press **R/S**.
5. To compute the interest rate paid on the savings element in the policy for that year, key in an assumed price per \$1000 for an alternative low-cost policy and press **R/S**.
6. To compute the true cost and interest rate for a different policy year or for a different policy, go to step 3.

Example:

Consider an average \$50,000 face amount cash value policy. The premium of \$1,010 is due at the beginning of the year, and a dividend of \$165 is received at the end of the policy year. A cash value of \$3,302 at the beginning of the year grows to \$4,104.

Assuming a 6% savings rate available elsewhere and that insurance protection could be purchased for \$3.00 per \$1,00, what is the true cost/\$1000 and the rate of return on your savings?

Keystrokes

Display

f CLEAR **ALL**

0.00

1010 **STO** 1

1,010.00

3302 **STO** 2

3,302.00

4104 **STO** 3

4,104.00

165 **STO** 4

165.00

50000 **STO** 5

50,000.00

6 **R/S**

6.57

\$, true yearly cost (YPT_t)

3 **R/S**

2.20

%, interest paid by policy

Real Estate

Equity Investment Analysis

Equity Investment Analysis is a method of evaluating income producing real estate investment alternatives on a pretax basis. Two key factors in this type of analysis are the anticipated income stream that the property will provide and the property's projected resale value at the end of the investment horizon. Based on this and the current price of the property, an equity yield rate can be found giving an indication of the profitability of the investment.

A brief explanation of terms frequently used in real estate analysis is given here in order to aid in understanding the problems and results more fully.*

Annual Net Cash Flow is the annual net operating income, without depreciation, minus the annual debt service (i.e., annual mortgage payments).

Reversion is the future sales price minus the mortgage balance at the end of the projection period.

Equity yield rate is that annual rate at which the present value of the net annual cash flows plus the present value of the equity reversion equals the equity investment value.

Equity investment value is the equity in the property at the beginning of the projection period.

Overall Capitalization Rate is the net operating income divided by the selling price.

Equity Yield Rate

Given the projection period in years, reversion amount, annual net cash flow, and equity investment value, the equity yield rate may be calculated as follows:

1. Set the Payment switch to END and press **f** CLEAR **FIN**.
2. Key in the reversion; press **FV**.
3. Key in the number of years projection; press **n**.

* For further information, refer to ELLWOOD TABLES, American Institute for Real Estate Appraisers, 1970.

26 Real Estate Equity Investment Analysis

4. Key in the net annual cash flow; press **PMT**.
5. Key in the equity investment value; press **CHS** **PV**.
6. Press **i** to obtain the equity yield rate.

Example :

An apartment complex is listed for \$490,000 and has an annual net operating income of \$41,600. The prospective buyer is considering a down payment of \$147,000 and will finance the remaining \$343,000 for 29 years at 8%. If the property appreciates a total of 20% over the next 10 years, what would the equity yield rate be?

Using calculations described in the Owner's Handbook, it is found that the monthly mortgage payments are \$2,538.01 and therefore the annual net cash flow is \$11,143.88 (the monthly payment is rounded).

(Net Operating Income—debt service = net cash flow)

The remaining mortgage balance (or reversion) at the end of 10 years will be \$297,017.73.

To calculate the reversion at the end of the tenth year, find the future sales price and subtract the remaining balance.

Keystrokes

Display

BEGIN  END

f CLEAR **FIN**

490000 **ENTER**

20 **%** **+**

588,000.00

Future sales price

297017.73 **-** **FV**

290,982.27

Reversion

10 **n** 11143.88 **PMT**

147000 **CHS** **PV** **i**

12.92

Equity yield rate

Equity Investment Value and Present Value

Given the desired equity yield rate, projection period, annual net cash flow, and the reversion, this procedure solves for the equity investment value and present value of the investment (current sales price). Information is entered as follows:

1. Set the Payment switch to END and press **f** CLEAR **FIN**.
2. Key in the reversion; press **FV**.

3. Key in the projection period in years; press **[n]**.
4. Key in the equity yield rate; press **[i]**.
5. Key in the annual net cash flow; press **[PMT]**.
6. Press **[PV]** for the equity investment value.
7. Key in the mortgage amount; press **[=]** to obtain the current sales price or present value.

Example :

An investor has some money he wants to invest in real estate. One of his alternatives is a warehouse, currently leased for 10 years, which generates \$26,460 annually before debt service (NOI). Because the warehouse is located in a growth area, he estimates the property should sell for \$420,000 at the end of 10 years. He can obtain an 8½%, 20 year mortgage for \$240,000 which would have monthly payments of \$2,082.78. If the desired yield is 11% over 10 years, what would his equity investment value be and how much could he pay for the property (what is the current sales price)?

Keystrokes

Display

BEGIN  END

[f] CLEAR **[FIN]**

10 **[9]** **[12x]** 8.5 **[9]** **[12÷]**

2082.78 **[CHS]** **[PMT]**

240000 **[PV]** **[FV]** **-167,984.38**

Remaining loan balance
after 10 years

420000 **[+]** **[FV]** **252,015.62**

Reversion value

10 **[n]** 11 **[i]**

[RCL] **[PMT]** 12 **[x]**

26460 **[+]** **[PMT]** **[PV]** **-97,393.37**

Equity investment value

240000 **[=]** **-337,393.37**

Current sales price

Future Sales Price and Overall Depreciation/ Appreciation Rate

This calculation solves for the sales price at the end of the projection period given the desired equity yield rate, annual net cash flow, equity investment value, projection period, and the mortgage balance at the end of the projection period. Information is entered as follows:

1. Set the Payment switch to END and press **f** CLEAR **FIN**.
2. Key in the projection period in years; press **n**.
3. Key in the equity yield rate; press **i**.
4. Key in the annual net cash flow; press **PMT**.
5. Key in the equity investment value; press **CHS** **PV**.
6. Press **FV** to compute the reversion amount.
7. Key in the mortgage balance at the end of the projection period and press **+** to obtain the required future sales price.
8. Key in the purchase price; press **x₂y** **f** **Δ%** to obtain the overall appreciation (if the answer is positive) or depreciation (if the answer is negative).

Example:

An apartment house has an annual net cash flow of \$14,211.24. The desired equity yield rate is 14% over a 9 year period. If the current asking price is \$616,000 what must the sales price at the end of year 9 be in order to achieve the desired 14% return? What overall appreciation does this represent?

(Assume 25% equity (\$154,000), 25 year mortgage at 8%, monthly payment of \$3,565.79, with a remaining balance of \$385,522.31 at the end of year 9).

Keystrokes

Display

BEGIN  END

f CLEAR **FIN**

9 **n** 14 **i**

14211.24	PMT		
154000	CHS	PV	FV
			272,207.35
385522.31	+		
			657,729.66
616000	x÷y	f	Δ%
			6.77
			Reversion
			Future sales price
			Overall appreciation

Homeowner's Monthly Payment Estimator

It is often useful, when comparison shopping for a mortgage or determining the appropriate price range of houses to consider, to be able to quickly estimate the monthly payment. This procedure calculates the approximate mortgage payment given the purchase price, tax rate per 1000, percent down, interest rate and term of the loan.

The calculation assumes that the assessed value is 100% of the sales price and does *not* take into account financing of the closing costs.

A simple keystroke procedure may be used to calculate the monthly payment:

1. Set the Payment switch to END and press **f** CLEAR **FIN**.
2. Key in the annual interest rate and press **9** **12÷**.
3. Key in the term of the loan (in years) and press **9** **12×**.
4. Key in the purchase price and press **STO** 1.
5. Key in the percent down and press **%** **-** **PV**.
6. Key in the tax rate in dollars per thousand, press **RCL** 1 **×** 12000 **÷** for approximate monthly taxes.
7. To calculate approximate monthly payments, press **ENTER** **+** **PMT** **PMT** **-** **CHS**. (Negative sign is the convention for cash paid out).

Example:

What would your monthly payments be on a \$39,000 house in a neighborhood with a \$25 per thousand tax rate and a 9 $\frac{3}{4}$ % interest rate on a 35 year loan with 10% down?

Keystrokes

BEGIN **f** END
f CLEAR **FIN**
 9.75 **9** **12÷**
 35 **9** **12×**
 39000 **STO** 1
 10 **%** **-** **PV**
 25 **ENTER** **+** **RCL** 1

Display

0.81
 420.00
 39,000.00
 35,100.00

Homeowner's Monthly Payment Estimator 31

x 12000 **÷**

81.25

(approximate monthly taxes)

ENTER+ **PMT** **PMT**

- **CHS**

-376.30

(approximate monthly payment)

The following short program may be used instead of the above.

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
9 P/R 9 CL P	00-	1	11- 1
RCL 1	01- 22 1	2	12- 2
RCL 2	02- 22 2	9 EE X	13- 25 32
%	03- 23	3	14- 3
-	04- 41	÷	15- 71
PV	05- 13	ENTER+	16- 31
ENTER+	06- 31	PMT	17- 14
9 LAST X	07- 25 31	PMT	18- 14
+	08- 51	-	19- 41
RCL 3	09- 22 3	CHS	20- 32
x	10- 61	9 P/R	

REGISTERS			
R ₀	R ₁ Purchase price	R ₂ % Down	R ₃ Tax rate

1. Key in the program.
2. Set payment switch to END, press **f** **CLEAR** **FIN**.
3. Key in the annual interest rate, press **9** **12÷**.
4. Key in the term of the loan in years, press **9** **12x**.
5. Key in the purchase price, press **STO** **1**.
6. Key in the percent down, press **STO** **2**.
7. Key in the tax rate in dollars per thousand, press **STO** **3**.
8. To calculate the approximate monthly payment, press **R/S**.

32 Homeowner's Monthly Payment Estimator

9. For a new case, store only the new variables by performing steps 3 thru 7 as needed. Press **[R/S]** for the new approximate monthly payment.

(Note that this program is especially useful on the HP-38C where Continuous Memory will retain the program and former variables which are still applicable).

Example:

Solve the previous example using the HP-38E/38C program.

Keystrokes

Display

BEGIN  END

[f] CLEAR **[FIN]**

9.75 **[9]** **[12÷]**

0.81

35 **[9]** **[12×**

420.00

39000 **[STO]** 1

39,000.00

10 **[STO]** 2

10.00

25 **[STO]** 3 **[R/S]**

-376.30

\$, approximate monthly
payment

What would the approximate payment be if the loan was at 10½% interest?

10.5 **[9]** **[12÷]** **[R/S]**

-396.50

\$, approximate monthly
payment

What if the down payment is increased to 20%?

20 **[STO]** 2 **[R/S]**

-361.47

\$, approximate monthly
payment

True Annual Percentage Interest Rate on a Mortgage With Fees

Borrowers are sometimes charged fees in connection with the issuance of a mortgage, which effectively raises the interest rate. The actual amount received by the borrower (PV) is reduced, while the periodic payments remain the same. Given the life or term of the mortgage, the interest rate, the mortgage amount, and the basis of the fee charge (how the fee is calculated), the true Annual Percentage Rate may be calculated. The following simple procedure may be used:

1. Set the Payment switch to END and press **[f]** CLEAR **[FIN]**.
2. Calculate and enter the periodic payment amount of the loan.
 - a. Key in the total number of payment periods; press **[n]**.
 - b. Key in the periodic interest rate; press **[i]**.
 - c. Key in the mortgage amount; press **[PV]**.*
 - d. To obtain the periodic payment amount press **[PMT]**.*
3. Calculate and key in the actual net amount disbursed.*
 - a. If fees are stated as a percentage of the mortgage amount (points), recall the mortgage amount (**[RCL]** **[PV]**); key in the fee (percentage) rate; press **[%]** **[-]** **[PV]**.
 - b. If fees are stated as a flat charge, recall the mortgage amount (**[RCL]** **[PV]**); key in the fee amount (flat charge); press **[-]** **[PV]**.
 - c. If fees are stated as a percentage of the mortgage amount plus a flat charge, recall the mortgage amount (**[RCL]** **[PV]**); key in the fee (percentage) rate, press **[%]** **[-]**; key in the fee amount (flat charge); press **[-]** **[PV]**.
4. Press **[i]** to obtain the percentage rate per compounding period.
5. To obtain the annual nominal percentage rate, key in the number of periods per year, and press **[x]**.

* Positive for cash received; negative for cash paid out.

34 True APR on a Mortgage With Fee

Example 1:

A borrower is charged 2 points for the issuance of his mortgage. If the mortgage amount is \$50,000 for 30 years, and the interest rate is 9% per year, with monthly payments, what annual percentage rate is the borrower paying? (1 point is equal to 1% of the mortgage amount.)

Keystrokes	Display	
BEGIN END		
CLEAR		
30	360.00	Months (into n)
9	0.75	% monthly interest rate (into i)
50000	-50,000.00	Loan amount (into PV)
	402.31	Monthly payment (calculated)
2	-49,000.00	Actual amount paid out by lender (into PV)
	0.77	% monthly interest rate (calculated)
12	9.23	Annual percentage rate

Example 2:

Using the same information as given in Example 1, calculate the APR if the mortgage fee is \$150 instead of a percentage.

Keystrokes	Display	
BEGIN END		
CLEAR		
30	360.00	Months (into n)
9	0.75	% monthly interest rate (into i)
50000	50,000.00	Loan amount (into PV)
	-402.31	Monthly payment (calculated)
150	49,850.00	Effective mortgage amount
	0.75	Monthly interest rate (calculated)
12	9.03	Annual percentage rate

Example 3:

Again using the information given in Example 1, what is the APR if the mortgage fee is stated as 2 points plus \$150?

Keystrokes**Display**BEGIN  END

f CLEAR FIN

30 9 12x

360.00

Months (into n)

9 9 12÷

0.75% monthly interest rate
(into i)

50000 PV

50,000.00

Loan amount (into PV)

PMT

-402.31Monthly payment
(calculated)

RCL PV 2 % -

150 - PV

48,850.00Effective mortgage
amount

i

0.77Monthly interest rate
(calculated)

12 x

9.26

Annual percentage rate

Rent Versus Buy

The question of whether to rent or buy a house is not always easy to answer, especially when the time period over which you would own or rent is short. To help make a decision, a financial analysis can be performed with your HP-38E/38C. The problem is best broken down into parts:

- I. Compute an estimated selling price and gross proceeds of the sale of a house after some period of time, assuming a reasonable appreciation rate.
- II. Compute the net proceeds from the sale of the house.
- III. Compute the yield on your investment in the house using net proceeds above for the future value.
- IV. Compute the value of a savings account where the initial deposit would be the same as the downpayment above, with monthly deposits amounting to the difference between mortgage payments and rent, compounded over the same time period as above, at a typical savings account rate.
- V. Compare the value of the hypothetical savings account to the net proceeds of the sale of the house, and compare the respective yields to arrive at your decision.

Steps I through IV may be performed using the following keystrokes:

- I.
 1. Set the payment switch at END and press **f** CLEAR **FIN**.
 2. Key in the time period in years and press **n**.
 3. Key in the annual appreciation rate of the property and press **i**.
 4. Key in the present value of the property and press **CHS** **PV**.
 5. Press **FV** to compute the future selling price.
 6. Key in the sales commission rate, press **%** **-** **STO** 0 to obtain gross proceeds of the sale.
- II.
 1. Press **f** CLEAR **FIN**.
 2. Key in the mortgage loan and press **PV**.
 3. Key in the periodic interest rate and press **i**.
 4. Key in the periodic payment amount and press **CHS** **PMT**.

5. Key in the number of periods and press **[n]** **[FV]** to obtain the mortgage balance.
 6. Press **[RCL]** **0** **[+]** **[FV]** to obtain the net proceeds from sale of the house.
- III.
1. Key in the down payment and closing costs, press **[CHS]** **[PV]**.
 2. Key in the monthly payments plus monthly cost of taxes, repairs and improvements and press **[ENTER+]**.
 3. Key in the rent which would otherwise be paid, press **[-]** **[CHS]** **[PMT]**.
 4. Key in the number of payment periods, press **[n]** **[i]** **12** **[x]** to obtain the yield on your housing investment.
- IV. Key in the annual interest rate of a typical savings account and press **[9]** **[12÷]** **[FV]** to calculate the value of an alternative bank account if you rented.

Example:

Assume that you are moving to Amherst, Massachusetts where you will be living for four years. Researching the housing in the area, you learn the following: Real estate is appreciating 15% annually; a house that would meet your needs costs \$40,000; with \$5000 down, you could get financing at 9.5% interest with monthly payments of \$301; taxes, repairs, and improvements would add another \$150 per month; closing costs would be about \$1000 on the purchase and 6% on the sale. Rent in the area runs around \$200 for suitable housing, and you could save the difference between house costs and rent in a special savings account that draws 7% interest. Compare the two investments.

Keystrokes**Display**I. **BEGIN**  **END****[f]** **CLEAR** **[FIN]**4 **[n]** **4.00**15 **[i]** **15.00**40000 **[CHS]** **[PV]** **-40,000.00****[FV]** **69,960.25**\$, market value after
4 years6 **[%]** **[-]** **[STO]** **0** **65,762.64**

less sales commission

II. **[f]** **CLEAR** **[FIN]**35000 **[PV]** **35,000.00**9.5 **[9]** **[12÷]** **0.79**301 **[CHS]** **[PMT]** **-301.00**

38 Rent Versus Buy

Keystroke:

Display:

48 **n** **FV** **STO** 1 **-33,610.02**
RCL 0 **+** **FV** **32,152.62**

mortgage balance
net proceeds from sale

III. 6000 **CHS** **PV** **-6,000.00**
451 **ENTER**
200 **-** **CHS** **PMT** **-251.00**
48 **n** **i** 12 **x** **20.70**

annual yield on investment

IV. 7 **g** **12÷** **FV** **21,789.84**

\$, value of bank account

V. Given the conditions in this example, purchasing a house would be a wise investment, realizing a greater net value and much higher return on investment (20.7% as compared to 7%).

At what average annual appreciation rate would the rent option be as attractive as purchasing the house?

Keystrokes

Display

f **CLEAR** **FIN** **21,789.84**

RCL 1 **CHS** **+**

1 **ENTER** 6 **%** **-**

÷ **FV** **58,936.02**

\$, market value (with 6% sales commission) to break even

4 **n**

40000 **CHS** **PV** **i** **10.17**

% annual rate of appreciation to break even

Appendix

Formulas

Tax Free Retirement Account (IRA) or Keogh Plan

n = the number of years to retirement

i = the compounded annual interest

PMT = the earnings used for investment (and taxes)

FV = future value

tax = the percent tax (expressed as a decimal)

For ordinary taxable investment:

$$FV = \frac{PMT}{i} [1 + i(1 - tax)] \{ [1 + i(1 - tax)]^n - 1 \}$$

For tax free investment:

$$FV = \frac{PMT}{i} (1 + i) [(1 + i)^n - 1]$$

Stock Portfolio Valuation and Analysis

n = the number of issues held

P_i = the current market price/share of a stock

S_i = the number of shares of a stock held

β_i = the beta coefficient of an individual stock

T = the total present value of a portfolio

Portfolio beta coefficient:

$$\beta = \sum_{i=1}^n \frac{P_i S_i \beta_i}{T}$$

U.S. Treasury Bill Valuation

$$\text{Price per \$100} = 100 - \left(\frac{\text{bid} + \text{ask}}{2} \right) \left(\frac{\text{days to maturity}}{360} \right)$$

True Annual Growth Rate of an Investment Portfolio

i_k = the k^{th} approximation of i , the time rate of return compounded continuously

P_j = the amount of the j^{th} transaction, positive for investments and negative for withdrawals.

e = the base of the natural logarithms

t_j = the date of the j^{th} transaction

V_T = the final fund value at date ‘‘T’’ (also the asset value or market value)

T = the date of which the rate of return is calculated

rate of return:

$$i_{k+1} = i_k + \frac{V_T - \sum P_j e^{i_k(T-t_j)}}{\sum (T-t_j) P_j e^{i_k(T-t_j)}}$$

Iteration continues until the difference between two successively computed values for the interest is less than 10^{-7} , $|i_{k+1}| < 10^{-7}$.

annual rate (i_{nominal}):

$$i_{\text{nominal}} = 100 \times \ln (i_{\text{continuous}} + 1).$$

Bonds

Reference:

Spence, Bruce M. and others, STANDARD SECURITIES CALCULATION METHODS, Securities Industry Association, 1973.

DIM/b = days between issue date and maturity date/day basis

DSM/b = days between settlement date and maturity date/day basis

DIS/b = DIM/b – DSM/b

DSC = E – DIS = days from settlement date to next six month coupon date

E = number of days in coupon period where settlement occurs

N = number of semi-annual coupons payable between settlement date and maturity date or call date

$$\text{CPN} = \frac{\text{CPN} \cdot \text{CALL}}{100}$$

CALL = redemption value per \$100 par value (call price)

Price (given yield) with more than 6 months to maturity.

$$\text{PRICE} = \frac{\text{CALL}}{\left(1 + \frac{\text{YIELD}}{200}\right)^{N-1 + \frac{\text{DSC}}{E}}} + \sum_{K=1}^N \frac{\frac{\overline{\text{CPN}}}{2}}{\left(1 + \frac{\text{YIELD}}{200}\right)^{K-1 + \frac{\text{DSC}}{E}}} - \left[\frac{\overline{\text{CPN}}}{2} \cdot \frac{\text{DIS}}{E} \right]$$

True Cost of an Insurance Policy

YPT_t = cost per \$1000 of protection in policy year t

P_t = annual premium for policy year t

V_t = cash value for policy year t

D_t = dividend for policy year t

F_t = face amount for policy year t

i = rate of return on savings element, expressed as a decimal (after-tax)

Cost per \$1000:

$$YPT_t = \frac{(P_t + V_{t-1})(1 + i) - (V_t + D_t)}{(.001)(F_t - V_t)}$$

Rate of return on savings:

$$i = \frac{(V_t + D_t) + (YPT_t)(.001)(F_t - V_t)}{(P_t + V_{t-1})} - 1$$

OTHER APPLICATIONS BOOKS WHICH ARE AVAILABLE

LENDING, SAVINGS, AND LEASING APPLICATIONS (00038-90025)

APR with Fees; Discounted Mortgages; Constant Principal Loans; Add-On Rate Converted to APR; Add-On Loan With Credit Life; Rule of 78's; Nominal Rate to Effective Rate; Number of Periods to Deplete a Savings Account; Periodic Deposits and Withdrawals; Savings Account Compounded Daily; Compounding Periods Different from Payment Periods; Advance Payments With Residual; Skipped Payments

INVESTMENT ANALYSIS AND STATISTICS APPLICATIONS FOR BUSINESS PROFESSIONALS AND STUDENTS (00038-90026)

Modified IRR (FMRR); Lease vs Purchase; Break-Even Analysis; Bonds; Exponential, Logarithmic and Power Curve Fits; Exponential Smoothing; Standard Error of the Mean; Grouped Data; Chi-Square; Normal Distribution; Covariance; Permutation; Combination; Random Number Generator

REAL ESTATE APPLICATIONS (00038-90024)

APR With Fees; Discounted Mortgages; Present Value and Yield of a Mortgage With Balloon Payment One Period After Last Payment; Deferred Annuities; Present Value of Increasing/Decreasing Annuity; Equity Yield Rate; Equity Investment Value and Present Value; Future Sales Price and Overall Depreciation/Appreciation Rate; Mortgage Constant; Refinancing; Wrap-Around Mortgages; Modified IRR (FMRR); Canadian Mortgages; Depreciation; Exponential Curve Fit

REAL ESTATE II: INCOME PROPERTY ANALYSIS APPLICATIONS (00038-90051)

Annual Property Cash Flow Analysis: Before-Tax Cash Flows and Reversions; After-Tax Cash Flows (including Multiple Mortgages); After-Tax Net Cash Proceeds of Resale. **Mortgage-Equity (Ellwood) Analysis:** Basic Rate and Overall Rate; Value (Present Worth) with R; Equity Dividend Rate; Cash Throw-Off to Equity; Value (Present Worth) with Dollar Amounts Given; Capital Appreciation or Depreciation on Resale; Equity Yield Rate from Dollar Figures. **Investment and Feasibility Analysis:** Feasibility Tests; Present Worth; Net Present Value; Profitability Index; Internal Rate of Return; Payback Period.

MARKETING AND FORECASTING APPLICATIONS (00038-90049)

Moving Average; Seasonal Variation Factors; Exponential Curve Fit; Gompertz Curve Fit; Forecasting with Exponential Smoothing; Breakeven Analysis; Operating Leverage; Profit and Loss Analysis; Markup and Margin Calculations; List and Net Prices; Learning Curve; Queuing and Waiting Line Theory; Cash Flow Loader; Percentage Tabulator.



Corvallis Division
1000 N.E. Circle Blvd., Corvallis, OR 97330, U.S.A.

00038-90052 Rev. B

Printed in U.S.A. 1/82